

CSC 460 Project 1

Spring 2019

Team 13

Hamzah Mansour	V00812108
Christina Araya	V00781339

1.0 Project Introduction

The objective of this project was to become familiarized with embedded system programming, hardware, and any hardware/software interfacing problems. This project also acts as preparation for producing a robot to play a laser game, which is yet to be specified, to be completed by the end of the term.

This project involves the following equipment:

- Two Arduino boards (ATmega2560)
- Two Bluetooth Modules
- Two PS2 Joysticks
- Two Servo Motors (pan and tilt)
- One Laser Pointer
- One Analog Light Sensor
- One LCD
- Two Breadboards
- One Power Supply

2.0 Phase 1

2.1 Description

This portion of the project required that a station be created using the materials provided excluding the Bluetooth modules. The station should control two servos, a laser, and the LCD shield as output. It should take the directions and button push of a PS2 controller joystick as input, along with the light detection provided by a light sensor. When the station is put together, the pan and tilt should be smoothly controllable using the joystick and the laser should be controlled by the push button on the joystick. The status of each directional axis of the joystick and the status of the push-button should be reflected on the LCD output, along with the status of the light sensor. If the laser is shone on the light sensor, this should be indicated on the LCD shield.

In order to properly wire the system a block diagram was used to understand how the inputs and outputs mapped to the arduino board. The block diagram can be found in Figure 1.




Figure 1: Phase 1 block diagram

The block diagram is colour coded by input and output associated with the various components. From the arduino board the colours are represented as following:

- Purple - light sensor component
- Red - laser component
- Black - Joystick X
- Blue - Push button
- Orange - Joystick Y
- Brown - Tilt servo
- Orange 2 - Pan servo
- Yellow - Ground
- Green - VCC

The pan and tilt kit is represented by two servos and the laser is represented by an led and resistor.

In order to regulate the speed at which the servo motors moved, a system was devised that used a timer to set the position of each servo. The tilt servo had a maximal tilt adjustment factor of 10 and the pan servo had a maximal tilt of 40 for greater speed, as it had a wider range of motion. They both sampled their relative speed from the x-axis and y-axis of the joystick. The speed of the servo was mapped from the position of the joystick to a range between negative and positive adjustment factors. The position excludes the dead band (440 - 570 for the y-axis and 425 and 575 for the x-axis) of the joystick and is linearly mapped. The speed is then limited to a smaller factor based on the previously sampled speed and added to a position which is limited to a range of 500 to 1000 for tilt and 750 and 2100 for pan. In this part a low pass filter was used on X, Y and button press (Z) inputs to reduce noise with an alpha value of 0.5. This was achieved by using exponential smoothing, as defined in equation 1 below:

$$(1) \quad s_t = \alpha x_t + (1 - \alpha)s_{t-1}, \quad t > 0, \quad 0 < \alpha < 1$$

This can then be written in a more simplified form show in equation 2:

$$(2) \quad s_t = \alpha x_t + (1 - \alpha)s_{t-1} = s_{t-1} + \alpha \cdot (x_t - s_{t-1})$$

With $\alpha=0.5$, this can be simplified as the following code:

```
y1 = y0 + ((y-y0)>>1);
```

Above the smoothed y1 value is computed by taking the old reading of y1, y0 and adding it to a weighted difference between the recently read y and the old value. A bitwise shift to the right (of y-y0) is equivalent to multiplying the difference by 0.5. However, z had an α -value of 0.3 to accommodate for its sensitivity.

In order to lessen the load on the servos, a timer was implemented to write the position if it had changed with the same frequency as in exercise 5, where the servo swept for a full period (360°) in about 5 seconds. However, the amount that the servo could move each update was bumped up depending on whether it was tilting (max 10 position change) or panning (max 40 position change). The position change was called the speed, in order to get the servos to update through the servo arduino library, the timers 1A,3A,4A or 5A couldn't be used and instead 3B and 4B were used and the timers were set in the interrupts instead of using the repeater scheduler that a (1,3,4 or 5) provided.

```
// in setup cli() at beginning and sei() at end
// timer setup for tilt
TCCR4A = 0;
TCCR4B = 0;
//Set prescaller to 256
TCCR4B |= (1<<CS42);

//Set TOP value (0.5 seconds)
OCR4B = 1736;
```

```

//Enable interrupt A for timer 3.
TIMSK4 |= (1<<OCIE4B);

TCNT4 = 0;

// interrupt handler
ISR(TIMER4_COMPB_vect){
    // routine actions here
    TCNT4 = 0;
}

```

Using the LiquidCrystal [1] Arduino library, with the LCD shield positioned on top of the arduino, the results from the light sensor output mentioned in the next paragraph, the outputs of x-axis, y-axis and push button from the thumbstick were displayed to the screen. In order to properly setup the LCD to output properly the pins for the LCD are set, and then started with set and begin functions. The LCD then will display when a cursor is set, and the print command is used. If the LCD shield is printed to with a high frequency, the output becomes difficult to read.

```

// select the pins used on the LCD panel
LiquidCrystal lcd(8, 9, 4, 5, 6, 7);

lcd.begin(16, 2);           // start the library
lcd.setCursor(0,0);
lcd.print("LGHT  JSX  JSY Z");

```

The photocell (light sensor) used was connected to the base station Arduino board as shown in fig. 1, with a resistor of 6.3K. The photocell was then covered with a sealed red plastic cup to diffuse incoming light. The resistance provided by the photocell was then measured to be around 160,000Ω and 170,000Ω with the ambient light in the lab. So, a threshold of 100,000Ω was chosen. Any resistance below this threshold of 100,000Ω would indicate that a light source had targeted the photocell and this set a flag that was then used to output the state of the photocell on the LCD [2].

3.0 Phase 2

3.1 Description


With the addition of two bluetooth modules, the scenario created in phase 1 is recreated in phase 2. In this iteration however the servo motors and laser were separated onto a second arduino board. The base station arduino board should send joystick and button press

commands to handle the pan, tilt, and laser to the remote board. The base station board should also handle the light sensor and LCD output, while adding time-triggered scheduling (TTS) to ensure that tasks only happen as often as they need to (setting different periods for each tasks), while still reducing CPU utilization.

3.2 Discussion

The base station as shown below is set up in a similar fashion to phase 1 just without the pan and tilt kit and with a bluetooth chip. Each of the outputs and inputs in Figure 2 correspond to a colour specified as follows:

- Purple - light sensor component
- Black - Joystick X
- Orange - Button Press
- Orange - Joystick Y
- Yellow - Ground
- Green - VCC



fritzing

Figure 2: Phase 2 Base station block diagram

The Remote station was also set up similarly to phase 1 it with a bluetooth chip and missing the shield, joystick and light sensor. Each of the coloured wires in Figure 3 corresponding to the following inputs and outputs:

- Red - laser component
- Blue - Bluetooth chip
- Brown - Tilt servo
- Orange 2 - Pan servo
- Yellow - Ground
- Green - VCC




Figure 3 : Phase 2 Remote station block diagram

Bluetooth

In phase 2, communication over bluetooth was introduced, with one board taking user input and the other controlling the servo motors and lasers. The base station arduino board was taking input (integers ranging from 0 to 1023) from the joystick (x-axis, y-axis, and push-button) and applying a low-pass filter. These input values are then used to calculate the desired speed of the servo motors, which is then mapped to a range, between 0 and 1023, while ignoring the

deadband in the centre of the ranges. This range was then mapped to a smaller range of integers: 3-84 in the case of the x-axis and 1-22 in the case of the y-axis, 1|2 in the case of the push-button. These integers were used to send data over bluetooth, two bytes at a time, with the first byte specifying the type of input (1:laser, 1:x-axis, 2:y-axis) and the second byte being the mapped value mentioned.

Since two bytes were sent to indicate what input or range needed to be sampled, at some point the Base and Remote stations would become out of sync, sending incorrect bytes to the remote station which would result in glitchy, repetitive behaviour. In order to stop this from occurring byte 0 was reserved as a reset bit, sent periodically to clear the bluetooth buffer. This ensures that in case the receiving bluetooth chip gets out of sync, the system can correct itself. Another problem that was encountered was that since the speed was originally being filtered on the base station when latency was introduced or there were some discrepancies with the received bytes the speed adjustments would get skipped. Because of this the relative position of the pan and tilt pins would start to jitter or in some cases jump erratically. This would happen due to reset bits or lost bytes and it was decided that the smoothing of the speed ensuring that the difference was no more than 2 different from the previous recorded speed should be performed on the remote station.

TTS: Time-triggered Scheduler

Another improvement that was made was due to the polling of the joystick controls, light sensor checks, sending data over bluetooth, and all outputs to the LCD where all happening in a single loop function, with a delay of 50ms. In order to plan the execution of each of these activities, a time-triggered scheduler was implemented [3] and a task was made for each activity, with the scheduler executing these tasks at a period specified during setup. In addition to these tasks, the actions on the remote station of controlling the laser and servo motors, as well as reading the data being sent over bluetooth were changed to be controlled by a time-triggered scheduler, and the maximum periods for those tasks were also found. The periods that were chosen as the maximum period for each task that would still result in an overall responsive system. The findings for these maximum periods can be seen in the table below.

An issue that occurred while switching over to this time-triggered scheduling approach was the reset byte that was sent in order to ensure the base and remote station were properly synchronised was being sent too often, as it was initially placed in the idle task, which the TTS would call whenever there was idle time (no other tasks needed to run). Given the low CPU utilization due to the relatively large periods chosen and short runtime of each task, this idle task happened so often that the resulting amount of reset bytes sent overwhelmed the remote station causing erratic behaviour and delays. This was solved by creating a task for sending a single reset byte with a period of 2 seconds. This meant that the two stations could not be out of sync for more than 2 seconds, but the reset byte would not be sent too often.

In order to check the CPU utilization of the tasks performed by the base and remote stations, high and low bits were set to pins on the start and completion of tasks. These were connected from both stations to a logic analyser, in order to measure the frequency, length of each task and overall period of tasks on each station. These values would be used to calculate the CPU utilization and adjusted to find the longest possible periods to keep the system working smoothly. Since the logic analyser was connected to both arduino boards initially, the ground connection had only been connected to one arduino board, the one that had more tasks, the Base station. However, because of this the output for the two tasks and the Idle task were erratic and all the readings were the same. When the logic analyser ground was connected to the ground of the Remote station as well, the periods became consistent with the rates set in code and the CPU utilization could be calculated. Because of this, it is likely only possible to read pins into the logic analyser from two boards at a time, as it only has two ground pins.

The logic analyzer was used to observe the occurrences and runtimes and a capture of this data can be seen below in figure 4 and figure 5.




Figure 4: runtimes and occurrences of base station tasks




Figure 5: runtimes and occurrences of remote station tasks

Base station

Task	Width/Runtime (ms)	Occurrences	Total Runtime
Poll X-axis joystick input	0.12	12	1.44
Poll Y-axis joystick input	0.12	12	1.44
Poll joystick push-button input	0.065	12	0.78
Light sensor	0.21	1	0.21
LCD	11.53	1	11.53
Send bluetooth data	0.027	5	0.135
Reset byte over bluetooth	0.0003125	1	0.0003125

Overall total runtime = 15.535ms

Overall period = 1205.984ms

CPU Utilization = (Overall total runtime / Overall period) / 100 = 1.29 %

Longest period with which the system as a whole still performs smoothly:

- Poll X-axis joystick control input: Tmax = 10ms
- Poll Y-axis joystick control input: Tmax = 10ms
- Poll Z-axis joystick push button input: Tmax = 10ms
- Light sensor task = 120ms
- LCD task = 120ms
- Send bluetooth = 20ms
- Reset bluetooth = 2 sec

Remote station

Task	Width/Runtime (ms)
Poll tilt servo output	0.0063
Poll pan servo output	0.0181

Overall total runtime = 0.0181 ms

Overall period = 19.46 ms

CPU Utilization = (Overall total runtime / Overall period) / 100 = 0.095%

Longest period with which the system as a whole still performs smoothly:

- Tilt task T max = 20ms
- Pan task T max = 20ms

4.0 Conclusion

The aim of the project was to familiarise with Arduino hardware and software interfacing. With the help of the logic analysers, online sample code and Arduino code examples, among others, the project could be realised. It was helpful to print the relevant information using the `Serial.print` functionality and using a logic analyser to time the tasks and see any patterns in the signals being sent. Most of the difficulties arose when circuits partially disconnected, which was particularly confusing when it occurred after some functionality was successfully implemented and, due to disconnections, the system no longer performing in the same way. Other difficulties arose when trying to send data over bluetooth or implement the time-triggered schedule for each task. Difficulties also arose due to bluetooth connection, latency or sampling rate issues. For example, whenever the remote station *read bluetooth* task would be placed in a scheduled event rather than the idle loop it would increase latency, as the bluetooth information wasn't immediately read and updated.

5.0 References

[1] LiquidCrystal Library. Available online at: <https://www.arduino.cc/en/Reference/LiquidCrystal>

[2] Photocell Hookup Guide. Available online at:
<https://learn.sparkfun.com/tutorials/photocell-hookup-guide/all>

[3] Lab guide: Time-triggered scheduling. Available online at:
<https://nrqm.ca/mechatronics-lab-guide/lab-guide-time-triggered-scheduling/>

[4] Project 1 Code: Phase 1 and 2. Available online at:
<https://github.com/HamzahMansour/csc460>

Appendix A

phase1.ino

```
1. #include <Servo.h>
2. #include <LiquidCrystal.h>
3.
4. // select the pins used on the LCD panel
5. LiquidCrystal lcd(8, 9, 4, 5, 6, 7);
6. int lcd_key    = 0;
7. int adc_key_in = 0;
8.
9. double z1 = 1;
10. double z0;
11. int x1 = 514;
12. int y1 = 506;
13. int z, x0, x, y0, y;
14. int JS1Z = 22;
15. int JS1X = A8;
16. int JS1Y = A10;
17. int PAN = 53;
18. int TILT = 52;
19. int LZ = A7;
20. //tiltvars
21. int maxAdjustTilt = 10;
22. int tiltSpeed = 0;
23. static unsigned int tiltValue = 500;
24. //panvars
25. int maxAdjustPan = 40;
26. int panSpeed = 0;
27. static unsigned int panValue = 2100;
28. Servo panServo;
29. Servo tiltServo;
30.
31. // Light sensor and LED
32. const int LIGHT_PIN = A15; // Pin connected to voltage divider output
33. const int LED_PIN = 26; // Use built-in LED as dark indicator
34. // Measure the voltage at 5V and the actual resistance of your
35. // 47k resistor, and enter them below:
36. const float VCC = 4.98; // Measured voltage of Arduinio 5V line
37. const float R_DIV = 4660.0; // Measured resistance of 3.3k resistor
38. // Set this to the minimum resistance require to turn an LED on:
39. const float DARK_THRESHOLD = 130000.0;
40.
41. void setup() {
42.   cli();
43.   pinMode(JS1Z, INPUT_PULLUP);
```

```

44. pinMode(JS1X, INPUT);
45. pinMode(JS1Y, INPUT);
46. pinMode(LZ, OUTPUT);
47. Serial.begin(9600);
48. panServo.attach(PAN, 750, 2100);
49. tiltServo.attach(TILT, 500, 1000);
50. // timer setup for tilt
51. TCCR4A = 0;
52. TCCR4B = 0;
53. //Set prescaler to 256
54. TCCR4B |= (1<<CS42);
55.
56. //Set TOP value (0.5 seconds)
57. OCR4B = 1736;
58.
59. //Enable interrupt A for timer 3.
60. TIMSK4 |= (1<<OCIE4B);
61.
62. TCNT4 = 0;
63.
64. // timer setup for pan
65. // clear timer config
66. TCCR3A = 0;
67. TCCR3B = 0;
68. //Set prescaler to 256
69. TCCR3B |= (1<<CS32);
70.
71. //Set TOP value (0.5 seconds)
72. OCR3B = 1736;
73.
74. //Enable interrupt A for timer 3.
75. TIMSK3 |= (1<<OCIE3B);
76.
77. //Set timer to 0 (optional here).
78. TCNT3 = 0;
79. sei();
80.
81. // LCD things
82. lcd.begin(16, 2); // start the library
83. lcd.setCursor(0,0);
84. lcd.print("LGHT JSX JSY Z");
85.
86. // Light sensor and LED
87. pinMode(LIGHT_PIN, INPUT);
88. pinMode(LED_PIN, OUTPUT);
89.
90.}
91.// tilt handling

```

```

92./*
93.At this point tilt doesnt appear to be working I'm going to head home also
94.there was some inexplicable shange in our x axis range
95.*/
96.ISR(TIMER4_COMPB_vect){
97.  int tiltInput = y1;
98.  int oldSpeed = tiltSpeed;
99.  if(tiltInput > 570){
100.      tiltSpeed = map(tiltInput, 570, 1021, 0, -maxAdjustTilt);
101.  }
102.  else if(tiltInput < 440){
103.      tiltSpeed = map(tiltInput, 440, 0, 0, maxAdjustTilt);
104.  }
105.  else{
106.      tiltSpeed = 0;
107.  }
108.  if(oldSpeed - tiltSpeed > 2){
109.      tiltSpeed = oldSpeed - 2;
110.  }
111.  else if( oldSpeed - tiltSpeed < -2){
112.      tiltSpeed = oldSpeed + 2;
113.  }
114.  tiltValue += tiltSpeed;
115.  if(tiltValue > 1000)
116.      tiltValue = 1000;
117.  if(tiltValue < 500)
118.      tiltValue = 500;
119.
120.  if(tiltSpeed != 0)
121.      tiltServo.writeMicroseconds(tiltValue);
122.  TCNT4 = 0;
123. }
124. //pan handling
125. ISR(TIMER3_COMPB_vect){
126.  int panInput = x1;
127.  int oldSpeed = panSpeed;
128.  if(panInput > 575){
129.      panSpeed = map(panInput, 525, 1022, 0, -maxAdjustPan);
130.  }
131.  else if (panInput < 425){
132.      panSpeed = map(panInput, 425, 0, 0, maxAdjustPan);
133.  }
134.  else{
135.      panSpeed = 0;
136.  }
137.
138.  if(oldSpeed - panSpeed > 2){
139.      panSpeed = oldSpeed - 2;

```

```

140.     }
141.     else if( oldSpeed - panSpeed < -2){
142.         panSpeed = oldSpeed + 2;
143.     }
144.     panValue += panSpeed;
145.
146.     if(panValue > 2100)
147.         panValue = 2100;
148.     if(panValue < 750)
149.         panValue = 750;
150.
151.     if (panSpeed != 0)
152.         panServo.writeMicroseconds(panValue);
153.     TCNT3 = 0;
154. }
155.
156. void loop() {
157.     // lazer handling
158.     z0 = z1;
159.     z = digitalRead(JS1Z);
160.     z1 = z0 + ((z-z0)*0.3);
161.     z = (int)(z1 + 0.5);
162.     analogWrite(LZ, (int)(1-z)*200);
163.     // pan handling
164.     x0 = x1;
165.     x = analogRead(JS1X);
166.     x1 = x0 + ((x-x0)>>1);
167.     //tilt handling
168.     y0 = y1;
169.     y = analogRead(JS1Y);
170.     y1 = y0 + ((y-y0)>>1);
171.
172.     // LCD print joystick X,Y,Z
173.     lcd.setCursor(5,1);
174.     lcd.print("          "); //0000 0000 0");
175.     lcd.setCursor(5,1);
176.     char printfStr[16];
177.     sprintf(printfStr,"%4d %4d %d",x1, y1,(int)(z1 + 0.5));
178.     lcd.print(printfStr);
179.
180.     // Light sensor and LED
181.     // Read the ADC, and calculate voltage and resistance from it
182.     int lightADC = analogRead(LIGHT_PIN);
183.     if (lightADC > 0)
184.     {
185.         // Use the ADC reading to calculate voltage and resistance
186.         float lightV = lightADC * VCC / 1023.0;
187.         float lightR = R_DIV * (VCC / lightV - 1.0);

```



```

188.      // If resistance of photocell is greater than the dark
189.      // threshold setting, turn the LED on.
190.      if (lightR >= DARK_THRESHOLD){
191.          digitalWrite(LED_PIN, LOW);
192.          lcd.setCursor(0,1);
193.          lcd.print("SAFE");
194.      }
195.      else{
196.          digitalWrite(LED_PIN, HIGH);
197.          lcd.setCursor(0,1);
198.          lcd.print("SHOT");
199.      }
200.  }
201.
202.  }

```

Phase2

Base.ino

```

1. #include "Arduino.h"
2. #include "scheduler.h"
3. #include <LiquidCrystal.h>
4.
5. // select the pins used on the LCD panel
6. LiquidCrystal lcd(8, 9, 4, 5, 6, 7);
7. int lcd_key  = 0;
8. int adc_key_in  = 0;
9.
10. int x, y, z, x0, y0, x1=506, y1=514, z_prev=1;
11. double z0, z1=1;
12. int JS1X=A8, JS1Y=A9, JS1Z=22;
13. int panSpeed = 0, maxAdjustPan = 40;
14. int tiltSpeed = 0, maxAdjustTilt = 10;
15. int panByte=3, tiltByte=1;
16. boolean shot=false;
17.
18. const int LIGHT_PIN = A15;
19. const float VCC = 4.98;
20. const float R_DIV = 4660.0;
21. const float DARK_THRESHOLD = 100000.0;
22.
23. // task function for joystick z-axis input pin (polling)
24. void poll_z_task(){ // every 10 ms
25.     digitalWrite(30, HIGH);
26.     // laser handling
27.     z0 = z1;
28.     z = digitalRead(JS1Z);
29.     z1 = z0 + ((z-z0)*0.3);
30.     z = (int)(z1 + 0.5);

```

```

31.  if(z != z_prev){
32.      z_prev = z;
33.      if(z==1) writeTwo(1,1);
34.      else if(z==0) writeTwo(1,2);
35.  }
36.  digitalWrite(30, LOW);
37.}
38.
39.void poll_x_task(){ // every 20ms
40.  digitalWrite(31, HIGH);
41.  // pan handling (x: 1,3 -> 1,44)
42.  x0 = x1;
43.  x = analogRead(JS1X);
44.  x1 = x0 + ((x-x0)>>1);
45.  panHandle(x1);
46.  digitalWrite(31, LOW);
47.}
48.
49.void poll_y_task(){ // every 20ms
50.  digitalWrite(32, HIGH);
51.  //tilt handling
52.  y0 = y1;
53.  y = analogRead(JS1Y);
54.  y1 = y0 + ((y-y0)>>1);
55.  tiltHandle(y1);
56.  digitalWrite(32, LOW);
57.}
58.
59.void light_sensor_task(){ // every 50ms
60.  digitalWrite(33, HIGH);
61.  int lightADC = analogRead(LIGHT_PIN);
62.  int a=0;
63.  if (lightADC > 0) {
64.      // Use the ADC reading to calculate voltage and resistance
65.      float lightV = lightADC * VCC / 1023.0;
66.      float lightR = R_DIV * (VCC / lightV - 1.0);
67.      // If resistance of photocell is greater than the dark
68.      if (lightR >= DARK_THRESHOLD){
69.          if(!shot) shot=true;
70.      }
71.      else{
72.          if(shot) shot=false;
73.      }
74.  }
75.  digitalWrite(33, LOW);
76.}
77.
78.void lcd_display_task(){ // every 120ms

```

```

79.  digitalWrite(34, HIGH);
80.  // LCD print joystick X,Y,Z
81.  lcd.setCursor(0,1);
82.  lcd.print("          ");
83.  if(shot){
84.      lcd.setCursor(0,1);
85.      lcd.print("SAFE");
86.  } else {
87.      lcd.setCursor(0,1);
88.      lcd.print("SHOT");
89.  }
90.  lcd.setCursor(5,1);
91.  char printfStr[16];
92.  sprintf(printfStr,"%4d %4d %d",x1, y1,(int)(z1 + 0.5));
93.  lcd.print(printfStr);
94.  digitalWrite(34, LOW);
95. }
96.
97. void send_bluetooth_task(){ // every 50ms
98.     digitalWrite(34, HIGH);
99.     writeTwo(1,panByte);
100.    writeTwo(2,tiltByte);
101.    digitalWrite(34, LOW);
102.
103. }
104.
105. void poll_reset_bluetooth(){ // every 2 sec
106.     digitalWrite(34, HIGH);
107.     Serial1.write(0);
108.     digitalWrite(34, LOW);
109. }
110.
111. // idle task
112. void idle(uint32_t idle_period){
113.     // send bluetooth?
114. }
115.
116. void setup(){
117.     Scheduler_Init();
118.     Scheduler_StartTask(0, 2, poll_z_task);
119.     Scheduler_StartTask(10, 5, poll_x_task);
120.     Scheduler_StartTask(20, 5, poll_y_task);
121.     Scheduler_StartTask(30, 5, send_bluetooth_task);
122.     Scheduler_StartTask(40, 120, light_sensor_task);
123.     Scheduler_StartTask(50, 120, lcd_display_task);
124.     Scheduler_StartTask(60, 2000, poll_reset_bluetooth);
125.
126.     Serial.begin(9600);

```

```

127.   Serial1.begin(9600);
128.   pinMode(JS1Z, INPUT_PULLUP);
129.   pinMode(30, OUTPUT);
130.   pinMode(31, OUTPUT);
131.   pinMode(32, OUTPUT);
132.   pinMode(33, OUTPUT);
133.   pinMode(34, OUTPUT);
134.   for(int i=30; i<35; i++) digitalWrite(i, LOW);
135.
136.   // LCD things
137.   lcd.begin(16, 2);   // start the library
138.   lcd.setCursor(0,0);
139.   lcd.print("LGHT  JSX  JSY Z");
140.
141.   // Light sensor
142.   pinMode(LIGHT_PIN, INPUT);
143. }
144.
145. void writeTwo(int a, int b){
146.   Serial1.write(a);
147.   Serial1.write(b);
148. }
149.
150. void panHandle(int panInput){
151.   int oldSpeed = panSpeed;
152.   if(panInput > 540){
153.     panSpeed = map(panInput, 540, 1021, 0, -maxAdjustPan);
154.   }
155.   else if (panInput < 440){
156.     panSpeed = map(panInput, 440, 0, 0, maxAdjustPan);
157.   }
158.   else{
159.     panSpeed = 0;
160.   }
161.
162.   if(panSpeed != oldSpeed){
163.     // set a global variable or array to a value
164.     panByte = map(panSpeed, -maxAdjustPan, maxAdjustPan-1, 3, 84);
165.   }
166. }
167.
168. void tiltHandle(int tiltInput){
169.   int oldSpeed = tiltSpeed;
170.
171.   if(tiltInput > 555){
172.     tiltSpeed = map(tiltInput, 555, 1022, 0, maxAdjustTilt);
173.   }
174.   else if(tiltInput < 425){

```

```

175.     tiltSpeed = map(tiltInput, 425, 0, 0, -maxAdjustTilt);
176. }
177. else{
178.     tiltSpeed = 0;
179. }
180.
181. if(tiltSpeed != oldSpeed){
182.     // set a global variable or array to a value
183.     tiltByte = map(tiltSpeed, -maxAdjustTilt, maxAdjustTilt-1, 1, 22);
184. }
185. }
186.
187. void loop(){
188.     uint32_t idle_period = Scheduler_Dispatch();
189.     if(idle_period){
190.         idle(idle_period);
191.     }
192. }
193.
194. int main(){
195.     init();
196.     setup();
197.     for (;;) { loop(); }
198.     for (;;) ;
199.     return 0;
200. }

```

Remote.iso

```

1. #include "scheduler.h"
2. #include <Servo.h>
3. int command[2] = {0,0};
4. int indx = 0;
5. int LZ = A15, PAN = 53, TILT = 52;
6. int panValue = 2050;
7. int tiltValue = 950;
8. int maxAdjustTilt = 10;
9. int maxAdjustPan = 40;
10. int speedTilt = 0;
11. int oldTiltSpeed = 0;
12. int speedPan = 0;
13. int oldPanSpeed = 0;
14. int panResponce = 0;
15. int tiltResponce = 0;
16. int readPin = 32, tiltPin = 30, panPin = 31;
17. Servo panServo;
18. Servo tiltServo;
19.
20. void setup() {

```

```

21. // put your setup code here, to run once:
22. Serial.begin(9600);
23. Serial1.begin(9600);
24. pinMode(LZ,OUTPUT);
25. panServo.attach(PAN);
26. tiltServo.attach(TILT);
27. pinMode(readPin, OUTPUT);
28. digitalWrite(readPin, LOW);
29. pinMode(tiltPin, OUTPUT);
30. digitalWrite(tiltPin, LOW);
31. pinMode(panPin, OUTPUT);
32. digitalWrite(panPin, LOW);
33. Scheduler_Init();
34.
35. // Start task arguments are:
36. //      start offset in ms, period in ms, function callback
37.
38. Scheduler_StartTask(0, 5, tiltHandler);
39. Scheduler_StartTask(2, 5, panHandler);
40. analogWrite(LZ,0);
41.}
42.// handling tilt
43.void tiltHandler(){
44.  digitalWrite(tiltPin, HIGH);
45.  int tempT = speedTilt;
46.  if(oldTiltSpeed > tempT){
47.    tempT = oldTiltSpeed - 1;
48.  }
49.  else if(oldTiltSpeed < tempT){
50.    tempT = oldTiltSpeed + 1;
51.  }
52.  if(tiltValue + tempT > 1000 || tiltValue + tempT < 500){
53.    tempT = 0;
54.  }
55.
56.  oldTiltSpeed = tempT;
57.  if(tempT != 0) {
58.    tiltValue += tempT;
59.    if(tiltValue >= 1000)
60.      tiltValue = 1000;
61.    if(tiltValue <= 500)
62.      tiltValue = 500;
63.    tiltServo.writeMicroseconds(tiltValue);
64.  }
65.  digitalWrite(tiltPin, LOW);
66.
67.}
68.//pan handling

```

```

69. void panHandler(){
70.   digitalWrite(panPin, HIGH);
71.   int tempP = speedPan;
72.   if(oldPanSpeed - tempP > 2){
73.     tempP = oldPanSpeed - 2;
74.   }
75.   else if( oldPanSpeed - tempP < -2){
76.     tempP = oldPanSpeed + 2;
77.   }
78.   if(panValue + tempP > 2100 || panValue + tempP < 750){
79.     tempP = 0;
80.   }
81.   oldPanSpeed = tempP;
82.   if(tempP != 0) {
83.     panValue += tempP;
84.     if(panValue >= 2100)
85.       panValue = 2100;
86.     if(panValue <= 750)
87.       panValue = 750;
88.     panServo.writeMicroseconds(panValue);
89.   }
90.   digitalWrite(panPin, LOW);
91.
92. }
93.
94. void callFunction(){
95.   int v0 = command[0], v1 = command[1];
96.   switch(v0){
97.     case 1:
98.       switch(v1){
99.         case 1:
100.            analogWrite(LZ, 200);
101.            break;
102.         case 2:
103.            analogWrite(LZ, 0);
104.            break;
105.         case 3 ... 84:
106.            speedPan = map(v1, 3, 84, -maxAdjustPan, maxAdjustPan);
107.            break;
108.         default: return;
109.       }
110.       break;
111.     case 2:
112.       switch(v1){
113.         case 1 ... 22:
114.            speedTilt = map(v1, 1, 22, -maxAdjustTilt, maxAdjustTilt);
115.            break;
116.         default: return;

```

```

117.     }
118.     break;
119.     default: return;
120.
121. }
122. //Serial.write(v0);
123. //Serial.write(v1);
124. }
125.
126. void readSerial1(){
127.     digitalWrite(readPin, HIGH);
128.     // put your main code here, to run repeatedly:
129.     if(Serial1.available()){
130.         int readVal = Serial1.read();
131.         if(readVal == 0) {
132.             indx = 0;
133.             digitalWrite(readPin, LOW);
134.             return;
135.         }
136.         command[indx++] = readVal;
137.         if(indx > 1){
138.             callFunction();
139.             indx = 0;
140.         }
141.     }
142.     digitalWrite(readPin, LOW);
143. }
144.
145. // idle task
146. void idle(uint32_t idle_period)
147. {
148.     // this function can perform some low-priority task while the scheduler
    has nothing to run.
149.     // It should return before the idle period (measured in ms) has expired.
    For example, it
150.     // could sleep or respond to I/O.
151.     readSerial1();
152. }
153.
154. void loop() {
155.
156.     uint32_t idle_period = Scheduler_Dispatch();
157.     if (idle_period)
158.     {
159.         idle(idle_period);
160.     }
161. }
162.

```



```
163. int main()
164. {
165.     init();
166.     setup();
167.
168.     for (;;)
169.     {
170.         loop();
171.     }
172.     for (;;)
173.     return 0;
174. }
```